Writing Self-Modifying Perl

Spencer Tipping

December 26, 2010

Contents

1	Introduction	3
Ι	The Basics	4
2	A Big Quine 2.1 A basic quine 2.2 Reducing duplication 2.3 Using eval	5 5 6
3	Building the interface 3.1 Using an editor	8 9
4	Namespaces 4.1 Handling functions more usefully 4.2 Catching attribute creation 4.3 Putting it all together 4.4 Separating bootstrap code	12 12 13 15 17
5	Serialization 5.1 Fixing the EOF markers 5.2 Verifying serialization 5.2.1 Implementing the Fowler-Noll Vo hash 5.2.2 Fixing EOF markers again 5.2.3 Implementing the state function 5.2.4 Implementing the verify function 5.3 Save logic 5.4 code::main fixes 5.5 Final result	21 22 22 24 24 24 25 25 26
6	Adding a REPL 6.1 The data data type	30 31 31 31

	6.5 6.6	Taking it to the max: tab-completion	32 33
7	Som	e improvements	39
	7.1	Useful functions	39
	7.2	Making some functions internal	40
	7.3	Separate attributes for data types	42
		7.3.1 Factoring externalization	43
	7.4	Abstracting %data	44
		7.4.1 Dynamic execution	45
	7.5	Final result	45
II	Th	ne Fun Stuff	52
8	eval	backtraces	53
9	Arch	niving state	55
	9.1	Saving state	55
	9.2		
			56
	9.3	Loading state	56 56
10		Loading state	56
10	Clor	Loading state	56 58
10	Clor 10.1	Loading state	56 58 58
10	Clor 10.1 10.2	Loading state The hypothetically function ning and inheritance Tracking inheritability Extensions to serialize	56 58 58 60
10	Clor 10.1 10.2 10.3	Loading state The hypothetically function ning and inheritance Tracking inheritability Extensions to serialize The update-from function	56 58 58 60 62
10	Clor 10.1 10.2 10.3	Loading state The hypothetically function ning and inheritance Tracking inheritability Extensions to serialize	56 58 58 60

Introduction

I've gotten a lot of WTF's¹ about self-modifying Perl scripts. Rightfully so, too. There's no documentation (until now), the interface is opaque and not particularly portable, and they aren't even very human-readable when edited:

```
meta::define_form 'meta', sub {
    my ($name, $value) = @_;
    meta::eval_in($value, "meta::$name");
};
meta::meta('configure', <<'__25976e07665878d3fae18f050160343f');
# A function to configure transients. Transients can be used to store any number of
# different things, but one of the more common usages is type descriptors.
sub meta::configure {
    my ($datatype, %options) = @_;
    $transient{$_}{$datatype} = $options{$_}} for keys %options;
}
__25976e07665878d3fae18f050160343f
...</pre>
```

Despite these shortcomings, though, I think they're fairly useful (this guide is a self-modifying Perl file, in fact). At the end, you'll have a script that is functionally equivalent to the object script, which I use as the prototype for all of the other ones.²

Proceed only with fortitude, determination, and Perl v5.10.

¹http://www.osnews.com/story/19266/WTFs_m

²See http://github.com/spencertipping/perl-objects for the full source.

Part I The Basics

A Big Quine

At the core of things, a self-modifying Perl script is just a big quine. There are only two real differences:

- 1. Self-modifying Perl scripts print into their own files rather than to standard output.
- 2. They print modified versions of themselves, not the original source.

If we're going to write such a script, it's good to start with a simple quine.

2.1 A basic quine

Some languages make quine-writing easier than others. Perl actually makes it very simple. Here's one:

Listing 2.1 examples/quine

```
1 my $code = <<'EOF';</pre>
```

- 2 print 'my \$code = <<\'EOF\';', "\n", \$code, "EOF\n"; print \$code;</pre>
- 3 **EOF**
- print 'my \$code = <<\'EOF\';', "\n", \$code, "EOF\n"; print \$code;</pre>

The logic is fairly straightforward, though it may not look like it. We're quoting a bunch of stuff using <<'EOF', and storing that into a string. We then put the quoted content outside of the heredoc to let it execute. The duplication is necessary; we want to quote the content and then run it. The key is this line:

```
print 'my $code = <<\'EOF\';', "\n", $code, "EOF\n"; print $code;</pre>
```

This code prints the setup to define a new variable \$code and prints its existing content after that.

¹A "quine" being a program that prints its own source.

²The single-quoted heredoc form doesn't do any interpolation inside the document, which is ideal since we don't want to worry about escaping stuff.

³Later on I'll use eval to reduce the amount of duplication.

2.2 Reducing duplication

Listing 2.2 examples/quine-with-data

We don't want to write everything in our quine twice. Rather, we want to store most stuff just once and have a quine that scales well. The easiest way to do this is to use a hash to store the state, and serialize each key of the hash in the self-printing code. So instead of creating \$code, we'll create %data:

```
1 my %data;
         2 $data{code} = <<'EOF';</pre>
         3 print 'my %data;', "\n";
         4 print '$data{', $_, '} = <<\'EOF\';', "\n$data{$_}EOF\n" for keys %data;</pre>
           print $data{code};
         7 print 'my %data;', "\n";
         8 print '$data{', $_, '} = <<\'EOF\';', "\n$data{$_}EOF\n" for keys %data;</pre>
           print $data{code};
            This is a good start. Here's how to add attributes without duplication:
Listing 2.3 examples/quine-with-data-and-foo
         1 my %data;
         2 $data{foo} = <<'EOF';</pre>
         3 a string
         4 EOF
         5 $data{code} = <<'EOF';</pre>
         6 print 'my %data;', "\n";
           print '$data{', $_, '} = <<\'EOF\';', "\n$data{$_}EOF\n" for keys %data;</pre>
           print $data{code};
         9 EOF
        10 print 'my %data;', "\n";
        print '$data{', $_, '} = <<\'EOF\';', "\n$data{$_}EOF\n" for keys %data;</pre>
        12 print $data{code};
```

2.3 Using eval

The business about duplicating \$data{code} is easily remedied by just evaling \$data{code} at the end. This requires the eval section to be duplicated, but it's smaller than \$data{code}. Here's the quine with that transformation:⁴

```
Listing 2.4 examples/quine-with-data-and-eval

my %data;

2 $data{foo} = <<'EOF';
```

⁴Note that these quines might not actually print themselves identically due to hash-key ordering. This is fine; all of the keys are printed before we use them.

```
3  a string
4  EOF
5  $data{code} = <<'EOF';
6  print 'my %data;', "\n";
7  print '$data{', $_, '} = <<\'EOF\\';', "\n$data{$_}EOF\n" for keys %data;
8  print $data{bootstrap};
9  EOF
10  $data{bootstrap} = <<'EOF';
11  eval $data{code};
12  EOF
13  eval $data{code};</pre>
```

The advantage of this approach is that all we'll ever have to duplicate is eval \$data{code} and my %data;, which is fairly trivial. It's important that you understand what's going on here, since this idea is integral to everything going forward.

Building the interface

Now that we've got attribute storage working, let's build a command-line interface so that we don't have to edit these files by hand anymore. There are a couple of things that need to happen. First, we need to get these scripts to overwrite themselves instead of printing to standard output. Second, we need a way to get and set entries in %data. Starting with the quine from the last section, here's one way to go about it:

```
Listing 3.1 examples/cli-basic
        1 my %data;
        2 $data{cat} = <<'EOF';</pre>
        3 sub cat {
             print join "\n", @data{@_};
        6 EOF
           $data{set} = <<'EOF';
           sub set {
             $data{$_[0]} = join '', <STDIN>;
          }
       11 EOF
        12 $data{code} = <<'EOF';</pre>
       # Eval functions into existence:
       14 eval $data{cat};
       15 eval $data{set};
       17 # Run specified command:
       18 my $command = shift @ARGV;
        20
       21 # Save new state:
       22 open my $fh, '>', $0;
        23 print $fh 'my %data;', "\n";
```

```
24 print $fh '$data{', $_, '} = <<\'EOF\';', "\n$data{$_}EOF\n" for keys %data;</pre>
   print $fh $data{bootstrap};
26 close $fh;
   $data{bootstrap} = <<'EOF';</pre>
28
29 eval $data{code};
30 EOF
31 eval $data{code};
   Now we can modify its state:
   $ perl examples/cli-basic cat cat
   sub cat {
     print join "\n", @data{@_};
   $ perl examples/cli-basic set foo
   bar
   ^D
   $ perl examples/cli-basic cat foo
   $
```

Not bad for a first implementation. This is a very minimal self-modifying Perl file, though it's useless at this point. It also has some fairly serious deficiencies (other than being useless). I'll cover the serious problems later on, but first let's address the usability.

3.1 Using an editor

The first thing that would help this script be more useful is a function that let you edit things with a real text editor. Fortunately this isn't difficult:

```
$ cp examples/cli-basic temp
$ perl temp set edit
sub edit {
    my $filename = '/tmp/' . rand();
    open my $file, '>', $filename;
    print $file $data{$_[0]};
    close $file;

    system($ENV{EDITOR} || $ENV{VISUAL} || '/usr/bin/nano', $filename);
    open my $file, '<', $filename;
    $data{$_[0]} = join '', <$file>;
    close $file;
}
```

```
^D
$
```

It won't work yet though. The reason is that we aren't evaling edit yet; we need to manually edit the code section and insert this line:

```
eval $data{cat};
            eval $data{set};
            eval $data{edit};
                                         # <- insert this
               Now you can invoke a text editor on any defined attribute:<sup>1</sup>
            $ perl examples/cli-editor edit cat
            # hack away
               Here's the object at this point:
Listing 3.2 examples/cli-editor
         1 my %data;
         2 $data{cat} = <<'EOF';</pre>
            sub cat {
              print join "\n", @data{@_};
            }
            EOF
         6
            $data{set} = <<'EOF';</pre>
            sub set {
              $data{$_[0]} = join '', <STDIN>;
        10
        11 EOF
        12
            $data{edit} = <<'EOF';
            sub edit {
              my $filename = '/tmp/' . rand();
        14
              open my $file, '>', $filename;
        15
              print $file $data{$_[0]};
        16
              close $file;
        18
              system($ENV{EDITOR} || $ENV{VISUAL} || '/usr/bin/nano', $filename);
        19
        20
              open my $file, '<', $filename;</pre>
              $data{$_[0]} = join '', <$file>;
        22
              close $file;
        23
        24 }
```

¹Don't modify bootstrap or break the print code though! This will possibly nuke your object.

```
25 EOF
26 $data{code} = <<'EOF';</pre>
27 # Eval functions into existence:
28 eval $data{cat};
29 eval $data{set};
30 eval $data{edit};
32 # Run specified command:
33 my $command = shift @ARGV;
34 &$command(@ARGV);
36 # Save new state:
37 open my $fh, '>', $0;
38 print $fh 'my %data;', "\n";
39 print $fh '$data{', $_, '} = <<\'EOF\';', "\n$data{$_}EOF\n" for keys %data;</pre>
40 print $fh $data{bootstrap};
41 close $fh;
42 EOF
43 $data{bootstrap} = <<'EOF';</pre>
44 eval $data{code};
45 EOF
46 eval $data{code};
```

Namespaces

It's a bummer to have to add a new eval line for every function we want to define. We could merge all of the functions into a single hash key, but that's too easy.¹ More appropriate is to assign a type to each hash key. This can be encoded in the name. For example, we might convert the names like this:

```
set -> function::set
cat -> function::cat
edit -> function::edit
code -> code::main
```

For reasons that I'll explain in a moment, we no longer need bootstrap. The rules governing these types are:

- 1. When we see a new function:: key, evaluate its contents.
- 2. When we see a new code:: key, evaluate its contents.

Rule 2 is why we don't need bootstrap anymore. Now you've probably noticed that these rules do exactly the same thing – why are we differentiating between these types then? Two reasons. First, we need to make sure that functions are evaluated before the code section is evaluated (otherwise the functions won't exist when we need them). Second, it's because functions can be handled in a more useful way.

4.1 Handling functions more usefully

Remember how we had to write sub X { and } every time we wrote a function, despite the fact that the function name was identical to the name of the key in %data? That's fairly lame, and it could become misleading if the names ever weren't the same. We really should have the script handle this for us. So instead of writing the function signature, we would just write its body:

¹Aside from being a lame cop-out, it also limits extensibility, as I'll explain later.

```
# The body of 'cat':
print join "\n", @data{@_};
```

and infer its name from the key. Perl is helpful here by giving us first-class access to the symbol table:

```
Listing 4.1 snippets/create-function

1 sub create_function {
2 my ($name, $body) = @_;
3 *{$name} = eval "sub {\n$body\n}";
4 }
```

If we're going to handle functions this way, we need to change the rule for function:: keys:

When we see a new function:: key, call create_function on the key name (without the function:: part) and the value.

4.2 Catching attribute creation

We can't observe when a new key is added to %data as things are now. Fortunately this is easy to fix. Instead of writing lines that read $data{...} = ...$, we can write some functions that perform this assignment for us, and in the process we can handle any side-effects like function creation. Here's a naive implementation:

```
Listing 4.2 snippets/define-function-define-code

sub define_function {
    my ($name, $value) = @_;
    $data{$name} = $value;
    create_function $name, $value;
}

sub define_code {
    my ($name, $value) = @_;
    $data{$name} = $value;
}
```

Since we're always going to assign into %data, we can abstract that step out:

```
Listing 4.3 snippets/define-definer

sub define_definer {
    my ($name, $handler) = @_;
    *{$name} = sub {
    my ($name, $value) = @_;
    $data{$name} = $value;
    &$handler($name, $value);
}
```

```
8  }
9  define_definer 'define_function', \&create_function;
10  define_definer 'define_code', sub {
11    my ($name, $value) = @_;
12   eval $value;
13  };
```

To avoid the possibility of later collisions we should probably use a separate namespace for all of these functions, since really bad things happen if you inadvertently replace one. I use the meta:: namespace for this purpose in my scripts.

At this point we've got the foundation for namespace creation. This is actually used with few modifications in the Perl objects I use on a regular basis. Here's meta::define_form lifted from object:

The idea is the same as define_definer, but with a few extra lines. We stash the delegate in a %datatypes table for later reference. We also (redundantly, I notice) create a function in the meta:: package so that we can refer to it when defining other forms. This lets us copy the behavior of namespaces but still have them be separate. The third line that's different is chomp \$value, which is used because heredocs put an extra newline on the end of strings. meta::define_form has the same interface as define_definer:

```
Listing 4.5 snippets/meta-define-form-function-code

1 meta::define_form 'function', \&create_function;
2 meta::define_form 'code', sub {
3 my ($name, $value) = @_;
4 eval $value;
5 };
```

Attribute definitions look a little different than they did before. The two define_form calls above create the functions meta::function and meta::code, which will need to be called this way:

```
meta::function('cat', <<'EOF');</pre>
```

```
print join "\n", @data{@_};
EOF
meta::code('main', <<'EOF');
# No more eval statements!
# Run command
...
# Save stuff
...
EOF</pre>
```

Notice that we don't specify the full name of the attributes being created. meta::function('x', ...) creates a key called function::x; this was handled in the define_form logic.

4.3 Putting it all together

At this point we're all set to write another script. The overall structure is still basically the same even though each piece has changed a little:

```
Listing 4.6 examples/namespace-basic
         1 my %data;
           my %datatypes;
            sub meta::define_form {
              my ($namespace, $delegate) = @_;
              $datatypes{$namespace} = $delegate;
              *{"meta::${namespace}::implementation"} = $delegate;
              *{"meta::$namespace"} = sub {
                my ($name, $value) = @_;
                chomp $value;
                $data{"${namespace}::$name"} = $value;
        12
                $delegate->($name, $value);
        13
           }
        14
           meta::define_form 'function', sub {
              my ($name, $body) = @_;
              *{\text{name}} = eval "sub {\n\Sbody\n}";
        18
           meta::define_form 'code', sub {
              my ($name, $value) = @_;
              eval $value;
        21
           };
        22
        24 meta::function('cat', <<'EOF');</pre>
           print join "\n", @data{@_};
```

```
EOF
26
27
28 meta::code('main', <<'EOF');</pre>
   # Run specified command:
   my $command = shift @ARGV;
   &$command(@ARGV);
   # Save new state:
   open my $file, '>', $0;
34
35
   # Copy above bootstrapping logic:
   print $file <<'EOF2';</pre>
   my %data;
   my %datatypes;
   sub meta::define_form {
41
     my ($namespace, $delegate) = @_;
42
     $datatypes{$namespace} = $delegate;
43
     *{"meta::${namespace}::implementation"} = $delegate;
44
45
     *{"meta::$namespace"} = sub {
       my ($name, $value) = @_;
46
        chomp $value;
47
        $data{"${namespace}::$name"} = $value;
48
        $delegate->($name, $value);
49
50
     };
51
   }
   meta::define_form 'function', sub {
     my ($name, $body) = @_;
53
     *{\text{name}} = eval "sub {\n$body\n}";
55
   meta::define_form 'code', sub {
     my ($name, $value) = @_;
     eval $value;
58
   };
59
   EOF2
60
61
   # Serialize attributes (everything else before code):
62
   for (grep(!/^code::/, keys %data), grep(/^code::/, keys %data)) {
     my ($namespace, $name) = split /::/, $_, 2;
     print $file "meta::$namespace('$name', <<'EOF');\n$data{$_}\nEOF\n";</pre>
   }
66
   # Just for good measure:
   print $file "\n__END__";
70 close $file;
71 EOF
```

```
73 <u>__END__</u>
```

The most substantial changes were:

- 1. We're defining two hashes at the beginning, though we still just use %data.
- 2. We're using delegate functions to define attributes rather than assigning directly into %data.
- 3. Quoted values now get chomped. I've added another \n in the serialization logic to compensate for this.
- 4. The serialization logic is now order-specific; it puts code:: entries after other things.
- 5. The file now has an __END__ marker on it.

4.4 Separating bootstrap code

The bootstrap code is now large quoted string inside code::main, which isn't optimal. Better is to break it out into its own attribute. To do this, we'll need a new namespace that has no side-effect.² I'll call this namespace bootstrap::.

```
meta::define_form 'bootstrap', sub {};
```

There's a special member of the bootstrap:: namespace that contains the code in the beginning of the file:

```
meta::bootstrap('initialization', <<'EOF');
my %data;
my %datatypes;
...
EOF</pre>
```

This condenses code::main by a lot:

```
Listing 4.7 snippets/bootstrapped-code-main

1 meta::code('main', <<'EOF');

2 # Run specified command:

3 my $command = shift @ARGV;

4 &$command(@ARGV);

5

6 # Save new state:

7 open my $file, '>', $0;

8 print $file $data{'bootstrap::initialization'};
```

²We can't use code:: because then the code would be evaluated twice; once because it's printed directly, and again because of the eval in the code:: delegate.

```
9
           # Serialize attributes (everything else before code):
        for (grep(!/^code::/, keys %data), grep(/^code::/, keys %data)) {
             my ($namespace, $name) = split /::/, $_, 2;
        13
              print $file "meta::$namespace('$name', <<'EOF');\n$data{$_}\nEOF\n";</pre>
           }
        14
        15
           # Just for good measure:
           print $file "\n__END__";
        18 close $file;
               Here's the final product, after adding the set and edit functions from
           before:
Listing 4.8 examples/namespace-full
        1 my %data;
           my %datatypes;
           sub meta::define_form {
              my ($namespace, $delegate) = @_;
              $datatypes{$namespace} = $delegate;
              *{"meta::${namespace}::implementation"} = $delegate;
              *{"meta::$namespace"} = sub {
                my ($name, $value) = @_;
        10
                chomp $value;
                $data{"${namespace}::$name"} = $value;
                $delegate->($name, $value);
              };
           }
        14
        15 meta::define_form 'bootstrap', sub {};
           meta::define_form 'function', sub {
              my ($name, $body) = @_;
        18
              *{\text{name}} = \text{eval "sub } {\text{n}}
           };
        19
           meta::define_form 'code', sub {
        20
              my ($name, $value) = @_;
              eval $value;
        23
        24
        25 meta::bootstrap('initialization', <<'EOF');</pre>
           my %data;
           my %datatypes;
        27
        28
           sub meta::define_form {
              my ($namespace, $delegate) = @_;
```

```
$datatypes{$namespace} = $delegate;
31
     *{"meta::${namespace}::implementation"} = $delegate;
32
     *{"meta::$namespace"} = sub {
33
       my ($name, $value) = @_;
34
35
        chomp $value;
        $data{"${namespace}::$name"} = $value;
        $delegate->($name, $value);
37
38
39
   meta::define_form 'bootstrap', sub {};
   meta::define_form 'function', sub {
     my ($name, $body) = @_;
     *{\text{name}} = eval "sub {\n$body\n}";
43
  meta::define_form 'code', sub {
     my ($name, $value) = @_;
     eval $value;
47
48 };
   EOF
49
50
51 meta::function('cat', <<'EOF');</pre>
   print join "\n", @data{@_};
   EOF
54
   meta::function('set', <<'EOF');</pre>
   $data{$_[0]} = join '', <STDIN>;
   EOF
57
58
  meta::function('edit', <<'EOF');</pre>
   my $filename = '/tmp/' . rand();
   open my $file, '>', $filename;
   print $file $data{$_[0]};
   close $file;
   system($ENV{EDITOR} || $ENV{VISUAL} || '/usr/bin/nano', $filename);
65
66
   open my $file, '<', $filename;</pre>
   $data{$_[0]} = join '', <$file>;
   close $file;
70 EOF
71
72 meta::code('main', <<'EOF');</pre>
   # Run specified command:
74 my $command = shift @ARGV;
   &$command(@ARGV);
```

```
77 # Save new state:
78 open my $file, '>', $0;
79 print $file $data{'bootstrap::initialization'};
81
   # Serialize attributes (everything else before code):
   for (grep(!/^code::/, keys %data), grep(/^code::/, keys %data)) {
     my ($namespace, $name) = split /::/, $_, 2;
     print $file "meta::$namespace('$name', <<'EOF');\n$data{$_}\nEOF\n";</pre>
85
86
   # Just for good measure:
88 print $file "\n__END__";
89 close $file;
90 EOF
91
92 __END__
```

Serialization

Earlier I alluded to a glaring problem with these scripts as they stand. The issue is the EOF marker we've been using. Here's what happens if we put a line containing EOF into an attribute:

It's not hard to see what went wrong: temp now has an attribute definition that looks like this:

```
meta::function('bif', <<'EOF');
print <<'EOF';
uh-oh...
EOF</pre>
```

We need to come up with some end marker that isn't in the value being stored. For the moment let's use random numbers.¹

¹object implements a simple FNV-hash and uses the hash of the contents. I'll go over how to implement this a bit later.

5.1 Fixing the EOF markers

There isn't a particularly compelling reason to inline the serialization logic in code::main. Since we have a low-overhead way of defining functions, let's make a serialize function to return the state of a script as a string, along with a helper method serialize_single to handle one attribute at a time:

Sorting the keys is important. We'll be verifying the output of the serialization function, so it needs to be stable.

Now code::main is a bit simpler. With these new functions the file logic becomes:

```
open my $file, '>', $0;
print $file serialize();
close $file;
```

5.2 Verifying serialization

What we've been doing is very unsafe. There isn't a backup file, so if the serialization goes wrong then we'll blindly nuke our original script. This is a big problem, so let's fix it. The new strategy will be to serialize to a temporary file, have that file generate a checksum, and make sure that the checksum is what we expect. Before we can implement such a mechanism, though, we'll need a string hash function.

5.2.1 Implementing the Fowler-Noll Vo hash

At its core, the FNV-1a hash² is just a multiply-xor in a loop. Generally it's written like this:

²http://en.wikipedia.org/wiki/Fowler-Noll-Vo_hash_function

In Perl it's advantageous to vectorize this function for performance reasons. It isn't necessarily sound to do this, but empirically the results seem reasonably well-distributed. Here's the function I ended up with:

```
Listing 5.3 snippets/fnv-hash-function
```

This produces a 32-bit hash. Ideally we have something of at least 128 bits, just to reduce the likelihood of collision. When I was writing the 128-bit hash I went a bit overboard with hash chaining (which doesn't matter because it isn't a cryptographic hash), but here's the full hash:

```
Listing 5.4 snippets/fast-hash-function
```

The convolutedness of this logic is partially to accommodate for very short strings.

5.2.2 Fixing EOF markers again

It's probably fine to use random numbers for EOF markers, but I prefer using a hash of the content. While it's probably about the same either way, it intuitively feels less likely that a string will contain its own hash.³

```
Listing 5.5 snippets/serialize-single-hash

1 meta::function('serialize_single', <<'EOF');

2 my ($namespace, $name) = split /::/, $_[0], 2;

3 my $marker = '__' . fast_hash($data{$_[0]});

4 "meta::$namespace('$name', <<'$marker'); \n$data{$_[0]} \n$marker";

5 EOF</pre>
```

We can also use the script state to get a tempfile in the edit function.⁴

5.2.3 Implementing the state function

The "state" of an object is just the hash of its serialization. (This is why it's useful to have the serialization logic factored out.)

5.2.4 Implementing the verify function

verify writes a temporary copy, checks its checksum, and returns 0 or 1 depending on whether the checksum came out invalid or valid, respectively. If invalid, it leaves the temporary file there for debugging purposes.

³And as we all know, intuition is key when making decisions in math and computer science...

⁴object uses File::Temp to get temporary filenames. This is a better solution than anything involving pseudorandom names in /tmp.

```
my $temporary_filename = "$0.$state";
open my $file, '>', $temporary_filename;
print $file $serialized_data;
close $file;
chmod 0700, $temporary_filename;

chomp(my $observed_state = join '', qx|perl '$temporary_filename' state|);

my $result = $observed_state eq $state;
unlink $temporary_filename if $result;
$result;
EOF
```

5.3 Save logic

Now we can use verify before overwriting \$0.

5.4 code::main fixes

There's actually a fairly serious problem at this point. Every script saves itself unconditionally, which involves creating a temporary filename and verifying its contents. What happens when we run one then? Something like this:

```
$ perl some-script cat function::cat
join "\n", @data{@_};  # Gets this much right
# Now calls save(), which calls verify() to create a new temp script:
```

```
> perl some-script.hash1 state
hash1  # Gets this much right
# Now calls save(), which calls verify() to create a new temp script:
> perl some-script.hash1.hash2 state
```

That's not what we want at all. There's no reason to call save unless a modification has occurred, so we can make this modification to code::main:

```
Listing 5.9 snippets/code-main-with-fixed-save

1 meta::code('main', <<'EOF');
2 my $initial_state = state();
3 my $command = shift @ARGV;
4 print &$command(@ARGV); # Also printing the result -- important for state
5 save() if state() ne $initial_state;
6 EOF</pre>
```

5.5 Final result

At this point we have an extensible and reasonably robust script. Here's what we've got so far:

```
Listing 5.10 examples/verified
         1 my %data;
         2 my %datatypes;
            sub meta::define_form {
         4
              my ($namespace, $delegate) = @_;
              $datatypes{$namespace} = $delegate;
              *{"meta::${namespace}::implementation"} = $delegate;
              *{"meta::$namespace"} = sub {
                my ($name, $value) = @_;
                 chomp $value;
                 $data{"${namespace}::$name"} = $value;
                 $delegate->($name, $value);
         12
              };
         13
         14 }
            meta::define_form 'bootstrap', sub {};
            meta::define_form 'function', sub {
              my ($name, $body) = @_;
         17
              *{\text{name}} = eval "sub {\n$body\n}";
         18
         19
         20 meta::define_form 'code', sub {
              my ($name, $value) = @_;
         21
              eval $value;
```

```
23 };
24
   meta::bootstrap('initialization', <<'EOF');</pre>
   my %data;
   my %datatypes;
27
28
   sub meta::define_form {
29
     my ($namespace, $delegate) = @_;
     $datatypes{$namespace} = $delegate;
     *{"meta::${namespace}::implementation"} = $delegate;
     *{"meta::$namespace"} = sub {
33
       my ($name, $value) = @_;
        chomp $value;
35
        $data{"${namespace}::$name"} = $value;
        $delegate->($name, $value);
37
38
     };
   }
39
   meta::define_form 'bootstrap', sub {};
   meta::define_form 'function', sub {
     my ($name, $body) = @_;
42
     *{\text{name}} = \text{eval "sub } {\text{n}} 
43
44
   };
   meta::define_form 'code', sub {
     my ($name, $value) = @_;
46
     eval $value;
   };
48
   EOF
49
50
   meta::function('serialize', <<'EOF');</pre>
   my @keys = sort keys %data;
   join "\n", $data{'bootstrap::initialization'},
               map(serialize_single($_), grep !/^code::/, @keys),
55
               map(serialize_single($_), grep /^code::/, @keys),
               "\n__END__";
56
   EOF
57
58
   meta::function('serialize_single', <<'EOF');</pre>
   my ($namespace, $name) = split /::/, $_[0], 2;
   my $marker = '__' . fast_hash($data{$_[0]});
   "meta::$namespace('$name', <<'$marker');\n$data{$_[0]}\n$marker";</pre>
63
65 meta::function('fnv_hash', <<'EOF');</pre>
66 my ($data) = @_{;}
67 my ($fnv_prime, $fnv_offset) = (16777619, 2166136261);
68 my $hash
                                  = $fnv_offset;
```

```
= 2 ** 32;
    my $modulus
    hash = (hash ^ ( _ & Oxffff) ^ ( _ >> 16)) * fnv_prime % modulus
      for unpack 'L*', $data . substr($data, -4) x 8;
    $hash;
    EOF
73
74
75 meta::function('fast_hash', <<'EOF');</pre>
    my ($data)
                   = @_;
    my $piece_size = length($data) >> 3;
    my @pieces
                   = (substr($data, $piece_size * 8) . length($data),
                      map(substr($data, $piece_size * $_, $piece_size), 0 .. 7));
    my @hashes
                   = (fnv_hash($pieces[0]));
    push @hashes, fnv_hash($pieces[$_ + 1] . $hashes[$_]) for 0 .. 7;
    hashes[$_] ^= hashes[$_+ 4] >> 16 | (hashes[$_+ 4] & 0xffff) << 16 for 0 .. 3;
    $hashes[0] ^= $hashes[8];
    sprintf '%08x' x 4, @hashes[0 .. 3];
    EOF
85
86
    meta::function('state', <<'EOF');</pre>
    fast_hash(serialize());
88
89
90
    meta::function('verify', <<'EOF');</pre>
    my $serialized_data = serialize();
    my $state
                        = state();
94
    my $temporary_filename = "$0.$state";
    open my $file, '>', $temporary_filename;
97     print $file $serialized_data;
98 close $file;
    chmod 0700, $temporary_filename;
    chomp(my $observed_state = join '', qx|perl '$temporary_filename' state|);
    my $result = $observed_state eq $state;
   unlink $temporary_filename if $result;
    $result;
103
    EOF
104
105
   meta::function('save', <<'EOF');</pre>
    if (verify()) {
      open my $file, '>', $0;
108
      print $file serialize();
109
      close $file;
110
   } else {
111
      warn 'Verification failed';
112
113 }
114 EOF
```

```
115
meta::function('cat', <<'EOF');</pre>
    join "\n", @data{@_};
118
    EOF
119
meta::function('set', <<'EOF');</pre>
    $data{$_[0]} = join '', <STDIN>;
    EOF
122
123
meta::function('edit', <<'EOF');</pre>
my $filename = '/tmp/' . rand();
open my $file, '>', $filename;
    print $file $data{$_[0]};
128 close $file;
    system($ENV{EDITOR} || $ENV{VISUAL} || '/usr/bin/nano', $filename);
    open my $file, '<', $filename;</pre>
    $data{$_[0]} = join '', <$file>;
   close $file;
132
133 EOF
134
135 meta::code('main', <<'EOF');</pre>
136 my $initial_state = state();
    my $command = shift @ARGV;
    print &$command(@ARGV);
    save() if state() ne $initial_state;
140 EOF
141
142 ___END___
```

Adding a REPL

There are some ergonomic problems with the script as it stands. First, it should have a shebang line so that we don't have to use perl explicitly. But more importantly, it should provide a REPL so that we don't have to keep calling it by name.

The first question is how this should be invoked. It would be cool if we could run the script without arguments and get the REPL, but that will require some changes to the current code::main. The "right way" to do it also requires a new data type.

6.1 The data data type

Sometimes we just want to store pieces of data without any particular meaning. We could use bootstrap:: for this, but it's cleaner to introduce a new data type altogether.

```
Listing 6.1 snippets/define-form-data

1 meta::define_form 'data', sub {
2     # Define a basic editing interface:
3     my ($name, $value) = @_;
4     *{$name} = sub {
5     my ($command, $value) = @_;
6     return $data{"data::$name"} unless @_;
7     $data{"data::$name"} = $value if $command eq '=';
8     };
9 };
```

This function we're defining lets us inspect and change a data attribute from the command line. Assuming data::foo, for example:

```
$ perl script foo = bar
bar
```

```
$ perl script foo
bar
$ perl script foo = baz
baz
$
```

6.2 Setting up the default action

The default action can be stored in a data:: attribute:

```
meta::data('default-action', <<'EOF');
shell
EOF

meta::code('main', <<'EOF');
...
my $command = shift @ARGV || $data{'data::default-action'};
print &$command(@ARGV);
...
EOF</pre>
```

Since all values are chomped already, we don't have to worry about the newline caused by the heredoc.

6.3 Making the script executable

This isn't hard at all. It means one extra line in the bootstrap logic, and another extra line in save:

```
meta::bootstrap('initialization', <<'EOF');
#!/usr/bin/perl
...
EOF

meta::function('save', <<'EOF');
...
    close $file;
    chmod 0744, $0; # Not perfect, but will fix later
...
EOF</pre>
```

6.4 The shell function

The idea here is to listen for commands from the user and simulate the @ARGV interaction pattern. Readline is the simplest way to go about this:

```
Listing 6.2
           snippets/shell-function-1
           meta::function('shell', <<'EOF');</pre>
           use Term::ReadLine;
         3 my $term = new Term::ReadLine "$0 shell";
           $term->ornaments(0);
           my $output = $term->OUT || \*STDOUT;
           while (defined($_ = $term->readline("$0\$ "))) {
             my @args = grep length, split /\s+|("[^"\\]*(?:\\.)?")/o;
             my $function_name = shift @args;
             s/^{"}(.*)"$/\1/o, s/\\\"/"/go for @args;
             if ($function_name) {
                chomp(my $result = eval {&$function_name(@args)});
                warn $@ if $@;
        13
               print $output $result, "\n" unless $@;
        14
        15
             }
           }
        16
           EOF
        17
```

This shell function does some minimal quotation-mark parsing so that you can use multi-word arguments, but otherwise it's fairly basic. The script's name is used as the shell prompt.

It's OK to use use inside of evaled functions. I think what happens is that it gets processed when the function is first created by meta::function. But basically, Perl does the right thing and it works just fine as long as the module exists.

6.5 Taking it to the max: tab-completion

If you have the GNU Readline library installed (Perl defaults to something else otherwise), you can get tab-autocompletion just like you can in Bash. Here's a complete function written by my wife Joyce, modified slightly to make sense with this implementation:

```
Listing 6.3 snippets/complete-function-1
    meta::function('complete', <<'EOF');
    my @attributes = sort keys %data;

sub match {
    my ($text, @options) = @_;
    my @matches = sort grep /^$text/, @options;

if (@matches == 0) {return undef;}
    elsif (@matches == 1) {return $matches [0];}
    elsif (@matches > 1) {
```

```
return ((longest ($matches [0], $matches [@matches - 1])), @matches);
11
     }
12
   }
13
14
   sub longest {
15
     my (\$s1, \$s2) = @\_;
     return substr ($s1, 0, length $1) if ($s1 ^ $s2) = ^(\0*)/;
17
     return '';
19
20
  my ($text, $line) = @_;
22 match ($text, @attributes);
   EOF
23
      Using this function is easy; we just add one line to shell:
   $term->Attribs->{attempted_completion_function} = \&complete;
   while (defined($_ = $term->readline("$0\$ "))) {
```

6.6 Final result

Merging the shell and executable behavior in with the script from the last chapter, we now have:¹

```
Listing 6.4 examples/shell
        #!/usr/bin/perl
        2 my %data;
           my %datatypes;
           sub meta::define_form {
             my ($namespace, $delegate) = @_;
             $datatypes{$namespace} = $delegate;
             *{"meta::${namespace}::implementation"} = $delegate;
             *{"meta::$namespace"} = sub {
        10
               my ($name, $value) = @_;
               chomp $value;
               $data{"${namespace}::$name"} = $value;
               $delegate->($name, $value);
        14
             };
           }
        15
           meta::define_form 'bootstrap', sub {};
           meta::define_form 'function', sub {
```

¹You might notice that I'm still using EOF as the marker in these scripts. As soon as the script is rewritten it will replace the EOFs with hashes; in general, you can use any valid delimiter the first time around and the script will take it from there.

```
my ($name, $body) = @_;
     *{\text{name}} = eval "sub {\n\Sbody\n}";
19
20
   meta::define_form 'code', sub {
     my ($name, $value) = @_;
     eval $value;
23
24 };
   meta::define_form 'data', sub {
     # Define a basic editing interface:
     my ($name, $value) = @_;
27
     *{$name} = sub {
28
       my ($command, $value) = @_;
       return $data{"data::$name"} unless @_;
30
        $data{"data::$name"} = $value if $command eq '=';
     };
33
   };
34
   meta::bootstrap('initialization', <<'EOF');</pre>
   #!/usr/bin/perl
37
   my %data;
   my %datatypes;
38
39
   sub meta::define_form {
40
     my ($namespace, $delegate) = @_;
41
     $datatypes{$namespace} = $delegate;
42
43
     *{"meta::${namespace}::implementation"} = $delegate;
     *{"meta::$namespace"} = sub {
44
       my ($name, $value) = @_;
45
        chomp $value;
        $data{"${namespace}::$name"} = $value;
47
        $delegate->($name, $value);
     };
49
   }
50
   meta::define_form 'bootstrap', sub {};
   meta::define_form 'function', sub {
     my ($name, $body) = @_;
     *{\text{name}} = \text{eval "sub } {\text{n}} 
54
   };
55
   meta::define_form 'code', sub {
56
     my ($name, $value) = @_;
     eval $value;
58
59
   meta::define_form 'data', sub {
60
     # Define a basic editing interface:
61
     my ($name, $value) = @_;
62
     *{$name} = sub {
```

```
my ($command, $value) = @_;
        return $data{"data::$name"} unless @_;
        $data{"data::$name"} = $value if $command eq '=';
      };
67
    };
68
    EOF
69
70
    meta::data('default-action', <<'EOF');</pre>
72
    shell
    FOF
73
74
    meta::function('serialize', <<'EOF');</pre>
    my @keys = sort keys %data;
    join "\n", $data{'bootstrap::initialization'},
               map(serialize_single($_), grep !/^code::/, @keys),
79
               map(serialize_single($_), grep /^code::/, @keys),
               "\n__END__";
80
    EOF
81
82
    meta::function('serialize_single', <<'EOF');</pre>
83
    my ($namespace, $name) = split /::/, $_[0], 2;
    my $marker = '__' . fast_hash($data{$_[0]});
    "meta::$namespace('$name', <<'$marker');\n$data{$_[0]}\n$marker";</pre>
    EOF
87
88
   meta::function('fnv_hash', <<'EOF');</pre>
89
   my ( data ) = @_;
    my ($fnv_prime, $fnv_offset) = (16777619, 2166136261);
92 my $hash
                                  = $fnv_offset;
                                  = 2 ** 32:
    my $modulus
    hash = (hash ^ ( _ & Oxffff) ^ ( _ >> 16)) * fnv_prime % modulus
      for unpack 'L*', $data . substr($data, -4) x 8;
    $hash;
97
    EOF
    meta::function('fast_hash', <<'EOF');</pre>
   my ($data)
                   = @_;
    my $piece_size = length($data) >> 3;
    my @pieces
                   = (substr($data, $piece_size * 8) . length($data),
102
                       map(substr($data, $piece_size * $_, $piece_size), 0 .. 7));
103
                   = (fnv_hash($pieces[0]));
104
    my @hashes
    push @hashes, fnv_hash(pieces[$_+ 1] . hashes[$_]) for 0 .. 7;
    hashes[_] = hashes[_+ 4] >> 16 | (hashes[_+ 4] & 0xffff) << 16 for 0 .. 3;
106
    $hashes[0] ^= $hashes[8];
    sprintf '%08x' x 4, @hashes[0 .. 3];
    EOF
109
```

```
meta::function('state', <<'EOF');</pre>
    fast_hash(serialize());
113
114
meta::function('verify', <<'EOF');</pre>
my $serialized_data = serialize();
    my $state
117
                         = state();
118
   my $temporary_filename = "$0.$state";
119
   open my $file, '>', $temporary_filename;
    print $file $serialized_data;
122 close $file;
chmod 0700, $temporary_filename;
    chomp(my $observed_state = join '', qx|perl '$temporary_filename' state|);
    my $result = $observed_state eq $state;
    unlink $temporary_filename if $result;
    $result;
127
    EOF
128
129
130 meta::function('save', <<'EOF');</pre>
    if (verify()) {
      open my $file, '>', $0;
      print $file serialize();
134
      close $file;
      chmod 0744, $0;
135
136 } else {
      warn 'Verification failed';
137
138
   }
   EOF
139
   meta::function('cat', <<'EOF');</pre>
    join "\n", @data{@_};
    EOF
143
144
    meta::function('set', <<'EOF');</pre>
    $data{$_[0]} = join '', <STDIN>;
146
    EOF
147
148
    meta::function('complete', <<'EOF');</pre>
   my @attributes = sort keys %data;
    sub match {
      my ($text, @options) = @_;
152
      my @matches = sort grep /^$text/, @options;
154
      if
            (@matches == 0) {return undef;}
155
```

```
elsif (@matches == 1) {return $matches [0];}
156
      elsif (@matches > 1) {
        return ((longest ($matches [0], $matches [@matches - 1])), @matches);
158
      }
159
160 }
    sub longest {
161
      my (\$s1, \$s2) = @\_;
      return substr ($s1, 0, length $1) if ($s1 ^ $s2) = ^ (\0*)/;
163
      return '';
164
165
   }
   my ($text, $line) = @_;
    match ($text, @attributes);
168
   meta::function('shell', <<'EOF');</pre>
171
    use Term::ReadLine;
my $term = new Term::ReadLine "$0 shell";
    $term->ornaments(0);
174 my $output = $term->OUT || \*STDOUT;
    $term->Attribs->{attempted_completion_function} = \&complete;
    while (defined($_ = $term->readline("$0\$ "))) {
      my @args = grep length, split /\s+|("[^"\\]*(?:\\.)?")/o;
177
      my $function_name = shift @args;
178
      s/^{"}(.*)"$/\1/o, s/\\"/"/go for @args;
179
180
      if ($function_name) {
181
        chomp(my $result = eval {&$function_name(@args)});
182
        warn $@ if $@;
183
        print $output $result, "\n" unless $@;
184
      }
185
    }
186
    EOF
187
188
    meta::function('edit', <<'EOF');</pre>
189
    my $filename = '/tmp/' . rand();
190
    open my $file, '>', $filename;
191
192 print $file $data{$_[0]};
   close $file;
    system($ENV{EDITOR} || $ENV{VISUAL} || '/usr/bin/nano', $filename);
    open my $file, '<', $filename;
    $data{$_[0]} = join '', <$file>;
    close $file;
197
    EOF
198
199
200 meta::code('main', <<'EOF');</pre>
   my $initial_state = state();
```

```
202 my $command = shift @ARGV || $data{'data::default-action'};
203 print &$command(@ARGV);
204 save() if state() ne $initial_state;
205 EOF
206
207 __END__
```

Some improvements

Let's step back for a minute and improve things a bit in preparation for some real awesomeness. There are few places that could use improvement. First, there isn't a way to get a list of defined attributes on an object without opening it by hand. Second, the interface exposes too many functions to the user; in particular, things like complete aren't useful from the command line. Finally, every data type we define gets put into bootstrap::initialization, which causes O(n) redundancy in the size of the data type constructors.

7.1 Useful functions

The most important thing to add is 1s, which gives you a listing of attributes: Related are cp and rm, which do what you would expect:

¹object contains a much more sophisticated version of 1s. It parses options and applies filters to the listing, much like the UNIX 1s command. I'll go over how to implement this stuff in a later chapter.

Another useful function is create, which opens an editor for a new attribute:²

5 **EOF**

Now we can create stuff from inside the shell or command-line and have a civilized text-editor interface to do it.

7.2 Making some functions internal

It would be nice to have a distinction between functions meant for public consumption and functions used just inside the script. For example, nobody's going to call fnv_hash from the command-line; they'd have to pass it a string in @ARGV, which isn't practical. So it's time for a new toplevel mechanism, the %externalized_functions table:

```
# In bootstrap::initialization:
my %data;
my %externalized_functions;
my %datatypes;
```

%externalized_functions maps every callable function to the attribute that defines it, and only the listed functions will be usable directly from the shell or the command-line. This has an additional benefit of providing much better autocompletion, since the first word in the REPL always names a function.

```
meta::define_form 'data', sub {
   my ($name, $value) = @_;
   $externalized_functions{$name} = "data::$name";
   *{$name} = ...;
};

meta::define_form 'function', sub {
   my ($name, $value) = @_;
   $externalized_functions{$name} = "function::$name";
   *{$name} = ...;
};
```

And here's the new data type:

²We can already do this with edit, but object doesn't let you edit attributes that don't exist. I'll include that behavior in these scripts before too long.

```
Listing 7.3 snippets/internal-function-type
        1 meta::define_form 'internal_function', sub {
             my ($name, $value) = @_;
             *{\text{ne}} = eval "sub {\n}";
        4 };
              We can now move fnv_hash, fast_hash, and complete into this namespace.
              We'll need to update shell and complete to leverage this new information:
Listing 7.4
           snippets/shell-2
        meta::function('shell', <<'EOF');</pre>
        use Term::ReadLine;
        3 my $term = new Term::ReadLine "$0 shell";
        4 $term->ornaments(0);
        5 my $output = $term->OUT || \*STDOUT;
           $term->Attribs->{attempted_completion_function} = \&complete;
           while (defined($_ = $term->readline("$0\$ "))) {
             my @args = grep length, split /\s+|("[^"\\]*(?:\\.)?")/o;
             my $function_name = shift @args;
             s/^{"}(.*)"$/\1/o, s/\\"/"/go for @args;
             if ($function_name) {
        12
               if ($externalized_functions{$function_name}) {
                 chomp(my $result = eval {&$function_name(@args)});
        14
                 warn $@ if $@:
        15
                 print $output $result, "\n" unless $@;
        16
                 warn "Command not found: '$function_name' (use 'ls' to see available commands)";
        18
        19
             }
        20
        21 }
        22 EOF
Listing 7.5 snippets/complete-2
        meta::function('complete', <<'EOF');</pre>
        2 my @functions = sort keys %externalized_functions;
        3 my @attributes = sort keys %data;
           sub match {
             my ($text, @options) = @_;
             my @matches = sort grep /^$text/, @options;
                    (@matches == 0) {return undef;}
             elsif (@matches == 1) {return $matches [0];}
             elsif (@matches > 1) {
               return ((longest ($matches [0], $matches [@matches - 1])), @matches);
        10
        11
             }
        12 }
```

```
sub longest {
     my (\$s1, \$s2) = @\_;
     return substr ($s1, 0, length $1) if ($s1 ^ $s2) = \( \^(\0*)/; \)
     return '';
16
17
18 my ($text, $line) = @_;
  if ($line = \(^{/}\) {
     # Start matching attribute names.
     match ($text, @attributes);
21
  } else {
     # Start of line, so it's a function.
     match ($text, @functions);
25
   }
26 EOF
```

7.3 Separate attributes for data types

It's cumbersome to have all of the data types go in bootstrap::initialization. Better is to break the code into separate attributes. To do this we'll need to restructure the scripts a little bit.

Up until now the "stuff first, code second" approach has worked out all right. But now we want to evaluate stuff at the beginning and at the end, and if this keeps up it could get out of hand. Better is to have serialize generate a call into some function that will be defined, and do away with code:: altogether. We can use a new namespace meta:: for stuff that needs to be evaluated at the beginning. So basically, instead of this:

```
grep(/^meta::/, @keys),
grep(!/^meta::/, @keys)),
"internal::main();",
"__END__";
```

And here's the definition for meta:: (it's identical to the one we used to have for code::). This is the only define_form invocation in bootstrap::initialization; the others now reside in their own attributes.

```
Listing 7.7 snippets/define-form-meta

1 meta::define_form 'meta', sub {
2    my ($name, $value) = @_;
3    eval $value;
4 };

   Here are the new type definitions:

meta::meta('type::data', <<'EOF');
meta::define_form 'data', sub {...};
EOF
meta::meta('type::function', <<'EOF');
meta::define_form 'function', sub {...};
EOF
meta::meta('type::bootstrap', <<'EOF');
meta::define_form 'bootstrap', sub {};
EOF
...</pre>
```

7.3.1 Factoring externalization

While we're cleaning up meta-stuff, it's worth thinking about factoring out externalization. There isn't a particularly good reason to keep manually assigning to %externalized_functions; better is to abstract this detail into a function. To do this, we'll want a meta-library:

```
Listing 7.8 snippets/meta-externalize

1 meta::meta('externalize', <<'EOF');
2 sub meta::externalize {
3 my ($name, $attribute, $implementation) = @_;
4 $externalized_functions{$name} = $attribute;
5 *{$name} = $implementation;
6 }
7 EOF</pre>
```

This meta-definition is available to the others because it sorts first.³ Now instead of manually externalizing stuff, data types like function:: and data:: can just use meta::externalize:

 $^{^3}$ Which is a horrible way to manage dependencies, but it's worked so far.

```
Listing 7.9 snippets/function-type-with-externalize

1 meta::meta('type::function', <<'EOF');
2 meta::define_form 'function', sub {
3 my ($name, $value) = @_;
4 meta::externalize $name, "function::$name", eval "sub {\n$value\n}";
5 };
6 EOF</pre>
```

7.4 Abstracting %data

Another issue worth fixing is that you can assign into %data arbitrarily, particularly in ways that end up breaking deserialization. For instance, nothing is stopping you from creating a key called foo::bar even though there isn't a namespace called foo:: This problem can be solved at the interface level (i.e. inside edit, set, and such), but it's probably more useful to go a step further and abstract all access to %data.

Rather than writing to %data, then, we'll use an internal function called associate; and to read from it we'll use retrieve. These two functions also benefit from a couple more to separate out namespace components. The namespace function gives you the base part, and the attribute function gives you the rest.

```
Listing 7.10 snippets/namespace-attribute-retrieve-associate-functions
            meta::internal_function('namespace', <<'EOF');</pre>
            my (name) = @_;
             $name = s/::.*$//;
            $name;
            meta::internal_function('attribute', <<'EOF');</pre>
          8 my ($name) = @_{;}
            $name = s/^[^:]*:://;
             $name;
            EOF
         11
            meta::internal_function('retrieve', <<'EOF');</pre>
            my @results = map defined $data{$_} ? $data{$_} : file::read($_), @_;
            wantarray ? @results : $results[0];
            EOF
         16
         17
            meta::internal_function('associate', <<'EOF');</pre>
```

⁴I can't remember why I thought retrieve was necessary when I wrote object. As far as I know it still isn't; it's just there for symmetry I think.

⁵All four of these functions are taken directly from object.

```
my ($name, $value, %options) = @_;
my $namespace = namespace($name);
die "Namespace $namespace does not exist" unless $datatypes{$namespace};
$data{$name} = $value;
execute($name) if $options{'execute'};
EOF
```

7.4.1 Dynamic execution

One problem with the way we've defined cp is that you'll have to close and reopen the shell to get new functions to take effect. This is because while we're assigning into %data, we're not calling the handler associated with the namespace. The simplest way to fix this is to dynamically invoke that handler:

```
Listing 7.11 snippets/execute-function
```

```
meta::internal_function('execute', <<'EOF');
my ($name, %options) = @_;
my $namespace = namespace($name);
eval {&{"meta::$namespace"}(attribute($name), retrieve($name))};
warn $@ if $@ && $options{'carp'};
EOF</pre>
```

associate is already hooked up to use this function; all you have to do is pass an extra option:

```
associate('function::foo', '...', execute => 1);
```

7.5 Final result

Integrating all of these improvements into the previous chapter's script yields this monumental piece of work:⁶

Listing 7.12 examples/some-improvements

```
#!/usr/bin/perl
my %data;
my %externalized_functions;
my %datatypes;

sub meta::define_form {
my ($namespace, $delegate) = @_;
$datatypes{$namespace} = $delegate;
*{"meta::${namespace}::implementation"} = $delegate;
```

⁶This is the last full listing I'll provide here. The remaining chapters cover the concepts required to get from here to object. At this point the stuff going on in object should more or less make sense, though you'll want to use 1s-a rather than 1s to get a full listing of attributes.

```
*{"meta::$namespace"} = sub {
10
       my ($name, $value) = @_;
        chomp $value;
        $data{"${namespace}::$name"} = $value;
        $delegate->($name, $value);
14
     };
15
   }
16
   meta::define_form 'meta', sub {
18
     my ($name, $value) = @_;
     eval $value;
20
21
22
   meta::meta('externalize', <<'EOF');</pre>
   sub meta::externalize {
     my ($name, $attribute, $implementation) = @_;
     $externalized_functions{$name} = $attribute;
     *{$name} = $implementation;
27
   }
28
   EOF
29
30
   meta::meta('type::bootstrap', <<'EOF');</pre>
   meta::define_form 'bootstrap', sub {};
   EOF
33
34
   meta::meta('type::function', <<'EOF');</pre>
35
   meta::define_form 'function', sub {
     my ($name, $body) = @_;
37
     meta::externalize $name, "function::$name", eval "sub {\n$body\n}";
   };
39
   EOF
40
41
   meta::meta('type::internal_function', <<'EOF');</pre>
   meta::define_form 'internal_function', sub {
     my ($name, $value) = @_;
     *{\text{ne}} = eval "sub {\n}";
45
   };
46
   EOF
47
48
   meta::meta('type::data', <<'EOF');</pre>
49
   meta::define_form 'data', sub {
50
     # Define a basic editing interface:
51
     my ($name, $value) = @_;
     meta::externalize $name, "data::$name", sub {
53
       my ($command, $value) = @_;
54
       return $data{"data::$name"} unless @_;
```

```
$data{"data::$name"} = $value if $command eq '=';
      };
57
58 };
    EOF
59
60
61 meta::bootstrap('initialization', <<'EOF');</pre>
62 #!/usr/bin/perl
    my %data;
    my %externalized_functions;
    my %datatypes;
    sub meta::define_form {
      my ($namespace, $delegate) = @_;
68
      $datatypes{$namespace} = $delegate;
69
      *{"meta::${namespace}::implementation"} = $delegate;
70
      *{"meta::$namespace"} = sub {
71
        my ($name, $value) = @_;
72
        chomp $value;
73
        $data{"${namespace}::$name"} = $value;
74
75
        $delegate->($name, $value);
      };
76
77 }
    meta::define_form 'meta', sub {
      my ($name, $value) = @_;
      eval $value;
81
82 };
    EOF
83
85 meta::data('default-action', <<'EOF');</pre>
    shell
    EOF
87
88
   meta::internal_function('namespace', <<'EOF');</pre>
    my ($name) = @_;
    $name = s/::.*$//;
91
    $name;
92
    EOF
93
94
95 meta::internal_function('attribute', <<'EOF');</pre>
96 my ($name) = @_;
    $name = s/^[^:]*:://;
    $name;
98
    EOF
99
100
101 meta::internal_function('retrieve', <<'EOF');</pre>
```

```
my @results = map defined $data{$_} ? $data{$_} : file::read($_), @_;
    wantarray ? @results : $results[0];
    EOF
104
105
106 meta::internal_function('associate', <<'EOF');</pre>
107 my ($name, $value, %options) = @_;
108 my $namespace = namespace($name);
    die "Namespace $namespace does not exist" unless $datatypes{$namespace};
    $data{$name} = $value;
    execute($name) if $options{'execute'};
112
meta::internal_function('execute', <<'EOF');</pre>
115 my (\$name, \%options) = @\_;
116 my $namespace = namespace($name);
    eval {&{"meta::$namespace"}(attribute($name), retrieve($name))};
    warn $@ if $@ && $options{'carp'};
    EOF
119
    meta::function('serialize', <<'EOF');</pre>
   my @keys = sort keys %data;
    join "\n", $data{'bootstrap::initialization'},
               map(serialize_single($_),
                 grep( /^meta::/, @keys),
                 grep(!/^meta::/, @keys)),
               "internal::main();",
                "__END__";
128
    EOF
129
130
    meta::function('serialize_single', <<'EOF');</pre>
131
    my ($namespace, $name) = split /::/, $_[0], 2;
    my $marker = '__' . fast_hash($data{$_[0]});
    "meta::$namespace('$name', <<'$marker');\n$data{$_[0]}\n$marker";</pre>
    EOF
135
136
meta::function('fnv_hash', <<'EOF');</pre>
138 my ($data) = @_;
   my ($fnv_prime, $fnv_offset) = (16777619, 2166136261);
140 my $hash
                                  = $fnv_offset;
                                  = 2 ** 32;
   my $modulus
    $hash = ($hash ^ ($_ & 0xffff) ^ ($_ >> 16)) * $fnv_prime % $modulus
      for unpack 'L*', $data . substr($data, -4) x 8;
    $hash;
144
145 EOF
146
147 meta::function('fast_hash', <<'EOF');</pre>
```

```
148 my ($data)
                    = @_;
    my $piece_size = length($data) >> 3;
    my @pieces
                    = (substr($data, $piece_size * 8) . length($data),
                       map(substr($data, $piece_size * $_, $piece_size), 0 .. 7));
    my @hashes
                    = (fnv_hash($pieces[0]));
152
    push @hashes, fnv_hash($pieces[$_ + 1] . $hashes[$_]) for 0 .. 7;
    hashes[\]^{=} hashes[\]^{+} 4] >> 16 | (hashes[\]^{+} 4] & 0xffff) << 16 for 0 . . 3;
    $hashes[0] ^= $hashes[8];
    sprintf '%08x' x 4, @hashes[0 .. 3];
156
    EOF
157
158
    meta::function('state', <<'EOF');</pre>
    fast_hash(serialize());
160
161
    meta::function('verify', <<'EOF');</pre>
163
    my $serialized_data = serialize();
    my $state
                         = state();
165
166
    my $temporary_filename = "$0.$state";
167
    open my $file, '>', $temporary_filename;
169 print $file $serialized_data;
   close $file;
    chmod 0700, $temporary_filename;
    chomp(my $observed_state = join '', qx|perl '$temporary_filename' state|);
    my $result = $observed_state eq $state;
    unlink $temporary_filename if $result;
    $result;
175
    EOF
176
177
    meta::function('save', <<'EOF');</pre>
    if (verify()) {
      open my $file, '>', $0;
180
      print $file serialize();
181
      close $file;
182
      chmod 0744, $0;
183
    } else {
184
      warn 'Verification failed';
185
186
    EOF
187
188
    meta::function('ls', <<'EOF');</pre>
189
    join "\n", sort keys %data;
190
191
192
193 meta::function('cp', <<'EOF');</pre>
```

```
associate($_[1], retrieve($_[0]));
    EOF
196
    meta::function('rm', <<'EOF');</pre>
197
    delete @data{@_};
198
    EOF
199
200
    meta::function('cat', <<'EOF');</pre>
    join "\n", @data{@_};
202
   EOF
203
204
    meta::function('create', <<'EOF');</pre>
    return edit($_[0]) if exists $data{$_[0]};
    associate($_[0], $_[1] || "# Attribute $_[0]");
    edit($_[0]);
208
209
    EOF
210
    meta::function('set', <<'EOF');</pre>
    $data{$_[0]} = join '', <STDIN>;
213
    EOF
214
215 meta::function('complete', <<'EOF');</pre>
    my @functions = sort keys %externalized_functions;
    my @attributes = sort keys %data;
217
    sub match {
219
      my ($text, @options) = @_;
      my @matches = sort grep /^$text/, @options;
             (@matches == 0) {return undef;}
221
      elsif (@matches == 1) {return $matches [0];}
222
      elsif (@matches > 1) {
223
        return ((longest ($matches [0], $matches [@matches - 1])), @matches);
      }
225
226 }
    sub longest {
227
      my (\$s1, \$s2) = @\_;
228
      return substr ($s1, 0, length $1) if ($s1 ^ $s2) = \(^(\0*)/;\)
      return '';
230
231 }
232 my ($text, $line) = @_;
   if ($line = \(^{/}\) {
      # Start matching attribute names.
234
      match ($text, @attributes);
235
236 } else {
      # Start of line, so it's a function.
      match ($text, @functions);
238
239 }
```

```
EOF
240
241
   meta::internal_function('shell', <<'EOF');</pre>
   use Term::ReadLine;
    my $term = new Term::ReadLine "$0 shell";
244
    $term->ornaments(0);
    my $output = $term->OUT || \*STDOUT;
    $term->Attribs->{attempted_completion_function} = \&complete;
    while (defined($_ = $term->readline("$0\$ "))) {
      my @args = grep length, split /\s+|("[^"\\]*(?:\\.)?")/o;
249
      my $function_name = shift @args;
250
      s/^{"}(.*)"$/\1/o, s/\\"/"/go for @args;
251
252
253
      if ($function_name) {
        if ($externalized_functions{$function_name}) {
254
255
          chomp(my $result = eval {&$function_name(@args)});
          warn $@ if $@;
256
          print $output $result, "\n" unless $@;
257
        } else {
259
          warn "Command not found: '$function_name' (use 'ls' to see available commands)";
        }
260
      }
261
    }
262
    EOF
263
   meta::function('edit', <<'EOF');</pre>
265
    my $filename = '/tmp/' . rand();
    open my $file, '>', $filename;
    print $file retrieve($_[0]);
    close $file;
    system($ENV{EDITOR} || $ENV{VISUAL} || '/usr/bin/nano', $filename);
270
    open my $file, '<', $filename;
    associate($_[0]}, join '', <$file>);
    close $file;
    EOF
274
275
    meta::internal_function('internal::main', <<'EOF');</pre>
    my $initial_state = state();
    my $command = shift @ARGV || retrieve('data::default-action');
    print &$command(@ARGV);
    save() if state() ne $initial_state;
280
    FOF
281
282
    internal::main();
283
284
   __END__
285
```

Part II The Fun Stuff

eval backtraces

Our script is fairly awesome so far. It prevents us from creating attributes in namespaces that don't exist, since that would cause incorrect serialization, it verifies before it saves, etc. But there's one problem. Take a look at the error messages we get:

```
$ perl examples/some-improvements
examples/some-improvements$ create foo::bar
Namespace foo does not exist at (eval 9) line 4.
examples/some-improvements$
```

If there's a problem in some attribute, we have no information about the location of the error other than "eval n" and the line number relative to that. object solves this problem:

```
$ object
object$ create foo::bar
  [error] Namespace foo does not exist at internal_function::associate line 4.
object$
```

The key is to wrap eval in such a way that we can later resolve the meaningless numbers into useful locations. And to do this, we're going to need to modify the bootstrap code again.

```
my %data;
my %externalized_functions;
my %datatypes;
my %locations;  # Maps eval-numbers to attribute names
```

There's a beautiful hack to handle the eval processing. Watch this (also in bootstrap::initialization):¹

¹It actually doesn't have to be inside the bootstrap code, but it doesn't change often and is useful to have around, so I decided to put it there to save time.

```
Listing 8.1 snippets/meta-eval-in

sub meta::eval_in {
    my ($what, $where) = @_;
    # Obtain next eval-number and alias it to the designated location
    @locations{eval('__FILE__') = ^ /\(eval (\d+)\)/\} = ($where); # <- step 1
    my $result = eval $what; # <- step 2
    $@ = s/\(eval \d+\)/$where/ if $@;
    warn $@ if $@;
    $result;
}</pre>
```

By evaling __FILE__, we get the current eval number. So the next one will be whatever we eval next. This means that in the shell sessions above, %locations contains a mapping from 9 to internal_function::associate. Here's the function that converts an eval index into an attribute name:

```
Listing 8.2 snippets/translate-backtrace-function
```

```
meta::internal_function('translate_backtrace', <<'EOF');
my ($trace) = @_;
$trace = s/\(eval (\d+)\)/$locations{$1 - 1}/g;
trace;
EOF</pre>
```

Notice that we're subtracting one. The eval number that triggered the error will be one greater than the one we stored.²

Now that we have this mechanism, we can go back and convert eval calls into meta::eval_in:

```
Listing 8.3 snippets/using-eval-in
```

```
meta::define_form 'function', sub {
    my ($name, $value) = @_;
    meta::externalize $name, "function::$name",
        meta::eval_in("sub {\n$value\n}", "function::$name");
};

meta::define_form 'internal_function', sub {
    my ($name, $value) = @_;
    *{$name} =
    meta::eval_in("sub {\n$value\n}", "internal_function::$name");
};
```

²Good API design would resolve this ahead-of-time rather than at lookup time. I haven't gotten around to changing it yet though.

Archiving state

Suppose you're about to do something risky with a script and you want to take a snapshot that you can restore to. You could copy into another file, but that's a brute-force approach and it requires you to exit the script's shell. Better is to have some kind of internal state management, and that's where explicit states come into play.

Remember that %data is just a variable; we can do all of the usual things with it. We can store a state by doing a partial serialization into an attribute, and we can restore from that state by evaling that attribute. To do this we're going to need another namespace.

```
Listing 9.1 snippets/state-type

1 meta::meta('type::state', <<'EOF');
2 # No action when a state is defined
3 meta::define_form 'state', \&meta::bootstrap::implementation;
4 EOF</pre>
```

9.1 Saving state

It's tempting to think that this code would do what we want:

```
# Won't work:
associate("state::$_[0]", serialize());
```

Unfortunately, serialize generates three things that we don't want. These are the bootstrap section at the beginning, the call to internal::main() at the end, and any attribute in the state:: namespace.¹ We'll need to write a separate function to serialize just what we want:

¹If some states contained others, the script size would grow exponentially in the number of states.

```
Listing 9.2 snippets/current-state-function

1    meta::function('current-state', <<'EOF');
2    my @valid_keys = grep ! /^state::/, sort keys %data;
3    my @ordered_keys = (grep(/^meta::/, @valid_keys), grep(! /^meta::/, @valid_keys));
4    join "\n", map serialize_single($_), @ordered_keys;
5    EOF

And here's a save-state function to automate the state creation process:

Listing 9.3   snippets/save-state-function

1    meta::function('save-state', <<'EOF');
2   my ($state_name) = @_;
3   associate("state::$state_name", &{'current-state'}());
4   EOF</pre>
```

9.2 Loading state

This is not as straightforward as saving state. Because we're modifying %data live, we have to be careful about what happens in the event that something goes wrong. We also don't want to have stray %data elements or externalized functions. The easiest way to defend against errors is to save the current state before applying a new one. Here's the implementation of load-state:

```
Listing 9.4 snippets/load-state-function

1 meta::function('load-state', <<'EOF');
2 my ($state_name) = @_;
3 my $state = retrieve("state::$state_name");
4 &{'save-state'}{'_'};  # Make a backup
5 delete $data{$_} for grep ! /^state::/, keys %data;
6 %externalized_functions = ();
7 eval($state);  # Apply the new state
8 warn $@ if $@;
9 verify();  # Make sure it worked
10 EOF</pre>
```

If the load failed for some reason, you can restore using load-state _. If it failed badly enough to bork your load-state function, then you have a problem.

9.3 The hypothetically function

Related to state management is a function called hypothetically, which lets you try something out and then revert. It's used internally to examine the state of a modified copy without actually committing changes.² Here's how it's defined:

```
Listing 9.5 snippets/hypothetically-function
        meta::internal_function('hypothetically', <<'EOF');</pre>
        2 my %data_backup = %data;
        3 my ($side_effect) = @_;
        4 my $return_value = eval {&$side_effect()};
        5 %data = %data_backup;
        6 die $@ if $@;
        7 $return_value;
        8 EOF
              You can use it like this:
           my $x = hypothetically(sub {
             associate('data::foo', '10');
             retrieve('data::foo');
           });
           my $y = retrieve('data::foo');
           \# now x eq '10' and <math>y is undef
```

²This is covered in chapter 10.

Cloning and inheritance

This is probably the single coolest thing about self-modifying Perl programs. You've probably had this looming feeling that propagating updated versions of functions between scripts was going to be a complete nightmare. For a long time this was indeed the case; I had shell scripts that copied attributes out of one script and into another. Luckily I got tired of doing things that way and came up with the inheritance mechanism that's used now.

Inheritance isn't as simple as copying all of the attributes from one script into another. Certain namespaces like data:: are script-specific, for instance. We'll need to have some way to keep track of which namespaces should be inherited.

Another issue is getting attributes from one script into another. My first implementation of inheritance retrieved each attribute individually. It used 1s and cat for the transfer, which involved O(n) runs of whichever script was being inherited from. Obviously it was really slow. O(n) runs of a function containing n functions means $O(n^2)$ total time, and Perl isn't blazingly fast at evaling functions. Later on I extended serialize to return a bundle of attributes that the child then evaled.

10.1 Tracking inheritability

We're going to need another toplevel field if we want to store data about data types. We can't use %data, since we don't really want to save it (whatever we're storing would be regenerated automatically anyway). What we really need is a way to store transient information:

```
my %data;
my %externalized_functions;
my %datatypes;
my %transient;
```

%transient does nothing except store stuff while the script is running, and all of its information is discarded when the script exits. It's basically just a temporary workspace where we can stash stuff.

We can now use %transient to store things about data types. For convenience let's define meta::configure to do this for us:

```
Listing 10.1 snippets/meta-configure
         1 meta::meta('configure', <<'EOF');</pre>
         sub meta::configure {
              my ($datatype, %options) = @_;
              $transient{$_}{$datatype} = $options{$_} for keys %options;
         5 }
         6 EOF
               Now we can add a configuration to each datatype we define:
            meta::meta('type::function', <<'EOF');</pre>
            meta::configure 'function', inherit => 1;
            meta::define_form 'function', ...;
            EOF
            meta::meta('type::data', <<'EOF');</pre>
            meta::configure 'data', inherit => 0;
            meta::define_form 'data', ...;
            EOF
            meta::meta('type::internal_function', <<'EOF');</pre>
            meta::configure 'internal_function', inherit => 1;
            EOF
            meta::meta('type::bootstrap', <<'EOF');</pre>
            meta::configure 'bootstrap', inherit => 1;
             . . .
            EOF
            meta::meta('type::state', <<'EOF');</pre>
            meta::configure 'state', inherit => 0;
```

EOF

¹For some reason I decided to store the keys in the odd order of option-namespace instead of the other way around. I'm still not sure why I did it this way, but it doesn't seem to cause problems.

10.2 Extensions to serialize

serialize needs to be able to give us a bundle of code to create just the attributes that should be inherited. While we're at it, it would also be nice if it handed us just the meta:: attributes and then just the non-meta:: attributes. This way we can make sure that the meta:: attributes didn't break anything and bail out early if they did.

None of this is particularly challenging, but given that we're going to invoke serialize externally we should probably fix the %options stuff. (The last thing we want to write is something like qx(\$script serialize partial 1 meta 1 inheritable 1)). What we need is an adapter that turns command-line options into Perl hashes.² Here's a function that uses Getopt-style parsing:

```
Listing 10.2 snippets/separate-options-function
```

```
meta::internal_function('separate_options', <<'EOF');</pre>
   # Things with one dash are short-form options, two dashes are long-form.
   # Characters after short-form are combined; so -auv4 becomes -a -u -v -4.
   # Also finds equivalences; so --foo=bar separates into $$options{'--foo'} eq 'bar'.
   # Stops processing at the -- option, and removes it. Everything after that
   # is considered to be an 'other' argument.
   # The only form not supported by this function is the short-form with argument.
   # To pass keyed arguments, you need to use long-form options.
   my @parseable;
   push @parseable, shift @_ until ! @_ or $_[0] eq '--';
   my @singles = grep /^-[^-]/, @parseable;
   my @longs = grep /^--/,
                                @parseable;
                                 @parseable;
   my @others = grep ! /^-/,
   my @singles = map /-(.{2,})/ ? map("-$_", split(//, $1)) : $_, @singles;
   my %options;
     $options{$1} = $2 for grep /^([^=]+)=(.*)$/, @longs;
   ++$options{$_}
                       for grep ! /=/, @singles, @longs;
19
   ({%options}, @others, @_);
   EOF
22
```

The output of this function is a reference to a hash of any keyword arguments (where short-form arguments are treated as increments) followed by any non-switch arguments (either because they came after -- or because they didn't start with a dash at all. For example, processing the arguments -xy --z=foo bar would yield ($\{-x => 1, -y => 1, --z => foo\}$, bar).

Given the ability to pipe options into serialize on the command-line, we

²OK, I'm making a jump here. Later it will become clearer why it's good to do it this way.

³Given the similarity, I don't remember why I didn't just use Getopt::Long for this stuff. I think I must have been having a NIH day.

just need to have it support a reasonably flexible selection interface. We'll later need to have 1s support the same options, so let's factor the key selector into its own function:

Listing 10.3 snippets/select-keys-function meta::internal_function('select_keys', <<'EOF');</pre> **my** %options = $@_{;}$ my \$criteria = \$options{'--criteria'} || \$options{'--namespace'} && "^\$options{'--namespace'}::" || '.'; grep /\$criteria/ && (! \$options{'-i'} || \$transient{inherit}{namespace(\$_)}) && (! \$options{'-I'} || ! \$transient{inherit}{namespace(\$_)}) && (! \$options{'-S'} || ! /^state::/o) && (! \$options{'-m'} || /^meta::/o) && (! \$options{'-M'} || ! /^meta::/o), **sort keys** %data;

This function takes the %options hash output by separate_options as input and returns a list of keys into %data. The somewhat odd logical structure of the grep predicate is just implication: "if \$options{'-i'} is set, then the key's namespace must be inheritable."

The new serialize function is fairly simple; most of the heavy lifting is already done:

```
Listing 10.4 snippets/serialize-final
```

9

EOF 10

```
meta::function('serialize', <<'EOF');</pre>
   my ($options, @criteria) = separate_options(@_);
  my $partial = $$options{'-p'};
  my $criteria = join '|', @criteria;
   my @attributes = map serialize_single($_),
     select_keys(%$options, '-m' => 1, '--criteria' => $criteria),
     select_keys(%$options, '-M' => 1, '--criteria' => $criteria);
   my @final_array = @{$partial ? \@attributes :
                                   [retrieve('bootstrap::initialization'),
                                    @attributes,
                                    'internal::main();', '',
                                    '__END__']};
   join "\n", @final_array;
14
   EOF
```

-m and -M select meta:: and non-meta:: attributes, respectively. We also provide criteria if the user has selected any. They're joined together with a pipe symbol because that forms a disjunction inside a regular expression (and criteria are regexps for attribute names). We also, somewhat importantly, have a -p switch to produce a partial serialization. This leaves off the bootstrap code and the internal::main() call. The only difference between this and current-state is that current-state also leaves out state:: attributes.

 $^{^4}$ This version of serialize also will do that for you if you pass the -S option.

10.3 The update-from function

Here's where things start to get interesting. update-from handles the case where you have two scripts base and child, and you want child to inherit stuff from base.⁵ Here's a basic implementation:

```
Listing 10.5 snippets/update-from-function
         1 meta::function('update-from', <<'EOF');</pre>
         my ($options, @targets) = separate_options(@_);
            my %options = %$options;
            @targets or die 'Must specify at least one target to update from';
            my $save_state = ! ($options{'-n'} || $options{'--no-save'});
                                 $options{'-f'} || $options{'--force'};
            mv $force
            &{'save-state'}('before-update') if $save_state;
            for my $target (@targets) {
              eval qx($target serialize -ipm);
              eval qx($target serialize -ipM);
              reload();
                             # We're about to define this
              unless (verify()) {
         13
                if ($force) {
         14
                  warn 'Verification failed, but keeping new state';
                } else {
         16
         17
                  warn "Verification failed after $target; reverting";
                  return &{'load-state'}('before-update') if $save_state;
         18
         19
              }
         20
            }
         21
         22
            EOF
```

If child has this function, you can update it this way (assuming you've set execute permissions):

```
$ ./child update-from ./base
$
```

If all goes well it will execute without failing. base will be run twice: once to grab inheritable meta::-attributes, again to grab inheritable other attributes.

The reload function just calls execute on each member of %data. It's there to make sure that the new attributes and the old attributes work well together. Here's the definition:

 $^{^{5}}$ "Stuff" is deliberately vague. Presumably we want to inherit every inheritable attribute though.

10.4 Managing parents

If you're using scripts as add-on modules, it gets tiring to issue n update-from commands when you're using n modules. Further, you have to type out the whole path of the module, which might be in a separate directory. Finally, attributes that later get deleted or renamed in the addon modules won't be cleaned up in the child script. The way we've got inheritance set up is woefully incomplete.

The missing piece is parent tracking. We need to keep track of two things:

- 1. Which scripts have I inherited from?
- 2. Which properties did I inherit from each one?

Fortunately this isn't too hard. We just need a new namespace and some new functions.

10.4.1 The parent:: namespace

We can use the parent:: namespace to answer both of the above questions. Each attribute will take its name from the path to a script (e.g. parent::/usr/bin/script1), and its contents will be a newline-separated string of the attributes inherited from that script. Here's the type definition:

```
Listing 10.7 snippets/parent-type
```

```
meta::meta('type::parent', <<'EOF');
meta::configure 'parent', inherit => 1; # Transitive parents (explained below)
meta::define_form 'parent', \&meta::bootstrap::implementation;
EOF
```

Now we need to update update-from to populate this namespace. Before we do, though, we need to define uniqueness.

10.4.2 Uniqueness

Let's suppose you've got three objects a, b, and c, and each inherits from the previous one.⁶ Then b's parent is a, and c's parent is b. It's important for c to know that a is also its parent. The reason is that otherwise you're forced to update b before c, since c inherits only from b.

It's ultimately for this reason that parent:: attributes get inherited. Inheritance is certainly transitive (if c inherits from b, which inherits from a, then c inherits from a). There are, however, some logistical matters to be dealt with. The most important one has to do with ordering.

It shouldn't matter in which order the parents are inherited from. This is an interesting requirement, because it means that the things an object inherits

⁶I'm presupposing that inheritance works automatically even though we haven't defined it quite yet.

from each of its parents form disjoint sets. The only way to pull this off is if each object keeps track of how it's different from its own parents. The attributes that the child has but the parent doesn't are considered unique.